
News 

US Training Days: Thursday 10th and Friday 11th November 
These training days will take place at the Holiday Inn Rosslyn at Key Bridge, 1900 N Fort 
Myer Drive, ARLINGTON, VA. Details for our next UK training days will be available in 
due course. 

We are also able to supply smaller ‘tailored’ training sessions to Spike2 and Signal users 
during visits to institutes or departments, or when attending scientific meetings. If you are 
interested in organizing a local training day for Spike2 or Signal users please contact 
sales@ced.co.uk. 

CED Video Tutorials 
Video tutorials are now available to demonstrate some of the features described in the 
email newsletter articles. Click the Show me how! link beside the article to display the 
related video tutorial. Please notify sales@ced.co.uk if you have any problems accessing 
the video content. 

Future meetings and events 
Society for Neuroscience 2011 
Booth 2238 - Walter E. Washington Convention Center, 
Washington, DC, USA 
November 12th – 16th 2011 

Australian Neuroscience Society 2012 
Jupiters Hotel, 
Gold Coast, QLD, Australia 
January 29th – February 1st 2012 

Latest versions of Spike2 and Signal 
Updates for Spike2 and Signal are available from the CED downloads page, or by 
clicking on the links in the table below. Demonstration versions of the latest software are 
also available.   

 Spike2 downloads Released Signal downloads Released 
Spike2 version 7.08         10/11 Signal version 5.04         10/11 

Spike2 version 6.16        03/11 Signal version 4.09         08/11 

Spike2 demo Signal demo 

Scripts Spotlight 
Artifact removal 

Artifacts, unwanted, usually non-physiological components of the signal, come in all 
shapes and sizes, depending on the nature of your experiment. Sometimes, they simply 
offend the eye but on occasion, they seriously impede your analysis of the data and it is 
necessary to do something to remove or reduce them. There are several options: 

Digital filtering can help if the spectral content of the artifact is very different from that of 
the signal of interest to you. 

If the artifact occurs repeatedly and has a consistent waveform while the physiological 
data varies, then we can generate a reference artifact signal by averaging and then 
subtract the reference signal from the original data. The HumRemove.s2s script uses this 
method to delete mains-related artifacts. 

Another common type of artifact is a click or glitch caused by an electrical stimulus or by 
other lab equipment switching on or off. The ArtRem.s2s script provides a method of 
reducing these unsightly glitches. It allows you to bracket an offending time range with 
cursors and then replace it by joining the start and end of the marked region with a 
straight line. If you have a trigger signal marking each artifact, you can also use this script 
to replace multiple artifacts in a single pass. There is an updated version with extra 
features for Spike2 version7 users, ArtRem6.s2s, which can be found in your 
Spike7\scripts folder and a more basic version compatible with older versions of the 
software which can be downloaded here. 

Users of Spike2 v7 can also try an alternative script FixGlitch.s2s, also available in 
the Spike7\scripts folder. This script allows you to replace glitches, not with a 
straight line, but with a trace of similar spectral content to the preceding and following 
data. In many cases this patch will repair the data seamlessly leaving little or no trace of 
the artifact. The patch is created using linear prediction. It is a technique that might be 
used to smooth over the scratches on an old vinyl record. This is an advanced 
mathematical method and there is no space here to explain the technicalities (fortunately 
for me!). If you would like to investigate the method in more detail or you would like to 
embed the method in a script of your own then check out Spike2 online help for the script 
function LinPred() 

If you have any questions please contact geoff@ced.co.uk 

 

 

Q. I would like to be able to make notes during the experiment and make these 
associated with events during the recording.    

A. A TextMark channel is a special type of Marker channel in Spike2 that can be 
used to enter notes and comments during a recording. The TextMark channel 
needs to be set-up in the Channels tab of the sampling configuration by adding a 
new channel and selecting TextMark from the Type drop-down list in the Channel 
parameters dialog. 

 
TextMark channel parameters 

Channel 30 in the data file is always reserved for the TextMark channel during 
sampling and the Source field should be set to Manual to allow the user to enter 
notes and comments on-line. 

Once a data file is sampling, or open ready to sample, the TextMark dialog can 
be activated by pressing Ctrl+T. When you want to add a note, click the Set 
Time button in the dialog, which saves the current time of the recording. You can 
then enter text in the dialog and set a corresponding Code (colour) for the marker. 

TextMark data can be added manually or read from external equipment via a serial line by selecting a COM port in the Source drop-down list of the Channel parameters dialog. Settings for Baud rate and Data can also be set depending on the external device settings

Clicking Save in the dialog will then add the TextMark to the data file at the time 
saved when you first clicked the Set Time button. 

 
Adding TextMark data on-line 

TextMarks are displayed as ‘post-it’ notes in channel 30. You can edit the marker 
data by double-clicking on it, view the associated comment by leaving the mouse 
pointer on the marker or change the Channel draw mode to show the Text or 
State with options to display the associated code as well. TextMarks can also be 
used to display Vertical markers, which can be extended over all of the 
channels in the file. 

 
TextMarks displayed in State mode 

 

 
Dialog for displaying Vertical markers 

  

 

 



 

 

Scripts: Spike2  

Q. I would like to apply an IIR filter to two waveform channels on-line. 

A. The script, OnlineMultiChanFilter.s2s, uses updated versions of the 
IIRxxx() family of commands which allow the same filter to be applied to a 
sequence of channels in the data file. The script generates a dialog from which 
you can select the type of filter to apply and a ‘source’ waveform channel to use 
as the first filtered channel. Once the filter has been generated you can then 
select up to three additional waveform channels to filter alongside the source 
waveform. The additional channels must be sampled at the same rate as the 
source channel and only channels with matching sample rates are available for 
selection. The filtered data from the selected channels is then written to new 
memory channels in the data file during sampling. This script requires Spike2 
version 7.08. 

 

 

 

 

Q. Is there any way to copy MultiClamp 700 settings directly to the sampling 
configuration to be used in clamping experiments?  

A. Signal supports the use of the MultiClamp 700 and AxoClamp 900A amplifiers for 
clamp experiments by communicating directly with the associated amplifier 
control software, which needs to be installed on the same PC. Settings and 
scaling information can then be read by Signal and applied to the sampling 
configuration and directly to the data file during recording. 

 If support for either the MultiClamp 700 or AxoClamp 900A was selected during 
the installation of Signal an additional button is enabled within the Clamp tab of 
the sampling configuration. You can use this to read the ADC ports used for 
telegraph scaling along with the amplifier settings directly from the amplifier 
control software and use these values to automatically set the clamping sets, 
channels, clamping modes and ADC and DAC port calibration values. The 
options available are: 

• Read channel sets for clamping – This option initializes the clamping 
sets with stimulus and response channels derived from the amplifier 
settings 

• Read clamping modes for sets – Retrieves the clamping mode used 
for each clamping set as defined by the amplifier command software 

• Read channel names, scaling and units – Set all waveform channels 
generated from the relevant ADC ports as defined by the amplifier 
telegraph settings 

• Read DAC scaling and units – Sets up the DAC specified for each 
clamp set using the external command sensitivity read from the amplifier 

An extra option, Read All – chans, mode, ADC & DAC will apply all of the 
above at once. As well as allowing quick and reliable setup, these options can 
also be used to check the current amplifier settings and make it easier to setup 
output pulses for use during recording. 

Full details on using the MultiClamp 700 and AxoClamp 900A are available in the 
Signal manual.  

Scripts: Signal  
Q. Is there any way to control older Magstim 200 series models that don’t have a 

serial line connection using Signal? 

A. The script, Ctrl200series.sgs, can be used to control older Magstim 200 
models that have been fitted with a parallel port modification to enable external 
control of the intensity settings. The script uses multiple states to control the 
1401 digital outputs. It creates a suitable sampling configuration and allows the 
user to specify the Magstim output intensities and interval between outputs for 

CED
The Clamp tab of the sampling configuration can be enabled and disabled from the Edit Preferences dialog


'$OnlineMultiChanFilter.s2s|Script for applying IIR filters to multiple waveform channels on-line.
'User selects the type of filter to apply and nominates a source waveform channel to use as the first
'channel to filter. The user can then select up to 3 additional channels to filter 
'(these must be sampled at the same rate as the source channel)
'The script then writes the filtered data to new memory channels in the file during sampling

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'CED 27/10/11


const lo:= 0.0002;		            'The minimum low pass setting for filters (as a fraction of the sample rate)
const hi:= 0.4998;		            'The maximum high pass setting for filters (as a fraction of the sample rate)

var data%;						    'Handle of new data file (Spike2 sets it to 0)
var sTime;						    'The last time we looked at the idle routine
var dataFile%,chan%;			
var filtName%,filtType%;		    'Variables returned from dialog holding filter name to use and type
var order%,notch%,reson%;		    'Variables holding filter settings
var rate,cFreq1,cFreq2,eFreq1,eFreq2,ripple,freq,nFreq,qFact;	'variables holding filter settings
var int;
var nyquist,minVal;
var numPoints%;
var ok%;
var v%,d%,f%,c%;		            'variables to hold dialog field return values
var filtCh%[4];
var type%;
var useFilt%;
var filtList$[4];					'Array to hold filter names
var typeList$[4];                   'Array to hold filter types
var filtMessage$;
var gFloat%[20];                    'global for floating window states
var ch%;
var waveChans%[32];
var chanNames$[32];
var chanNum$[32];
var useChans%[32];
var checkchans%[32];
var useCh%:= 2;
var use%;
var u%:=1;
var maxAddCh%:= 3;


var chans%[34];                      'Default channels to filter

var save[1][1];                         'Filter settings array (to be resized when number of channels and filter type set)


if App(-1)<708 then Message("This script needs at least Spike2 version 7.08 to run.\nRegistered version 7 users can download the latest\nversion from our website"); halt endif;	'Checks that the version of Spike2 is suitable


filtList$[0]:="Butterworth";        'List of filters
filtList$[1]:="Bessel";
filtList$[2]:="Chebyshev type 1";
filtList$[3]:="Chebyshev type 2";

typeList$[0]:="Low pass";           'List of filter types
typeList$[1]:="High pass";
typeList$[2]:="Band pass";
typeList$[3]:="Band stop";

HideAll();                       'Hide all unwanted bars
ToolbarVisible(1);				 'Make toolbar visible always
New%();							 'Set up new sampling window
DoToolbar();					 'Do the toolbar
RestoreAll();                    'Restore all bars
Halt;

func New%()							'New sampling window
var i%;
View(LogHandle());					'Make log view the current view
EditSelectAll();					'Select all text in log view
EditClear();						'Delete it
Window(0,80,100,100);				'Display it at the bottom of the screen
WindowVisible(1);					'Make it visible

if data%>0 then						'If there is already a data view open...
  View(data%);						'...close it
  FileClose();                      'No point setting data% to 0...
endif;

data%:=FileNew(0,1);				'Open a new data file for sampling
if data%<0 then Message("Unable to open new data file");Halt() endif;
DrawMode(-1,2);						'Set event draw mode to lines
Window(0,0,100,80);					'Make data window in top bit of screen
XRange(0,10);

FrontView(LogHandle());				'Bring the Log view to the front
FrontView(data%);					'Bring the data view to the front
u%:=1;

IIRSettings%();
ChanSelect(chan%,1);
ChanList(waveChans%[],1+32768);
 for f%:=1 to waveChans%[0] do
        if  1/BinSize(waveChans%[f%]) = rate then
        useChans%[u%]:=waveChans%[f%];
        chanNames$[u%]:= ChanTitle$(waveChans%[f%]);
        chanNum$[u%]:= Print$("%d ",waveChans%[f%]);
        u%+=1;     
        endif;
    next;
useChans%[0]:= u%-1;

if useChans%[0] > 0 then
repeat
    SetChans%();
    useCh%:=1;
    ArrConst(chans%[],0);
for use%:=1 to useChans%[0] do
    if checkchans%[use%-1] = 1 then
            chans%[useCh%+1] := useChans%[use%];
            useCh%+=1;
    endif
        next;
    if useCh%-1 > maxAddCh% then
        Message("You have selected more than the maximum allowed additional channels.\nSelect a maximum of 3 additional channels to filter");
        endif;
until useCh%-1 <= maxAddCh%;

chans%[0]:= useCh%;
    chans%[1]:= chan%;
else
    chans%[0]:= 1;
    chans%[1]:= chan%
endif;

useFilt%:=IIRMake%();
for i% := 1 to chans%[0] do
    filtCh%[i%-1] := MemChan(0,chans%[i%]);
    ChanTitle$(filtCh%[i%-1], ChanTitle$(chans%[i%])+"Filt");
    ChanShow(filtCh%[i%-1]);
    ChanColour(filtCh%[i%-1],1,16);
    ChanSelect(chans%[i%],1);
    ChanComment$(filtCh%[i%-1],filtMessage$);
next;

ToolbarEnable(3,0);					'Disable "Sample stop" button
ToolbarEnable(2,1);					'Disable "Sample stop" button
ToolbarText("Press SAMPLE START to commence sampling");
return 1;
end;


Func IIRSettings%()
repeat		            'If settings are not compatible, the dialog repeats
	DlgCreate("On-line IIR Filter settings");
	DlgGroup("Select Filter Type",1,1,47,4);
	DlgList(1,"",filtList$[],4,4,2);
	DlgList(2,"",typeList$[],4,28,2);
	DlgGroup("Source data",1,5.5,47,4);
	DlgGroup("Filter Characteristics",1,10,47,6);
	DlgCheck(3,"Notch",3,3);
	DlgCheck(4,"Resonator",3,4);
	DlgInteger(5,"Set Filter Order|For Bandpass and Bandstop filters, the Order used is twice this setting",1,10,22,11.2,1);
    DlgChan(6,"Source channel|This is the first channel you want to filter",1+512,20,6.5);
    DlgReal(7,"Edge Frequency 1 (Hz)",0.1,250000,22,12.4,0.1);
	DlgReal(8,"Edge Frequency 2 (Hz)",0.1,250000,22,13.6,0.1);
	DlgReal(9,"Ripple (dB)|This is the desired pass band ripple for type 1, or the cut in the stop band for type 2",1,100,22,14.8,1);
	DlgGroup("Notch and Resonator settings",1,16.5,47,3.5);
	DlgReal(10,"Notch Freq. (Hz)",0.1,250000,22,17.7,0.1);
	DlgReal(11,"q-Factor|The higher the q value, the narrower the notch. If unsure, try 100 as a starting value",10,1000,22,18.9,1);
	DlgAllow(1023,0,DlgSettings%);
	
		
	ok%:=DlgShow(filtName%,filtType%,notch%,reson%,order%,chan%,eFreq1,eFreq2,ripple,Freq,qFact);
	

	if ok%=0 then RestoreAll();halt;endif;		'If user presses cancel, quit script

	if ok%>0 then 				'If user presses OK
            rate:= 1/BinSize(chan%);		'Get sample rate of selected channel
			
		
		if d%=1 or f%=1 then	 'The following statements check for enabled fields, and only calculates frequencies for those
			nFreq:= freq/rate;	 'All unused fields have values set to lo as they are not used
			cFreq1:=lo;
			cFreq2:=lo;
		else;
		
		if d%=0 and f%=0 then
		cFreq1:= eFreq1/rate;	'Calculate used frequencies as a fraction of sample rate to check compatibility
		cFreq2:=lo;
			if v%=2 or v%=3 then
				cFreq2:= eFreq2/rate;	
				else
				cFreq2:=lo;
			endif;
		nFreq:=lo;
		endif;
		endif;
        
        nyquist:= rate/2;
        minVal:= rate/5000;
        
        if cFreq1 > hi or cFreq1 < lo or cFreq2 > hi or cFreq2 < lo or nFreq > hi or nFreq < lo then		'If outside the acceptable range
            if notch% or reson% = 1 then
                Message("The Notch frequency must be less than %01.2f (1/2 the sample rate)\nand greater than %f (1/5000 the sample rate) for the selected data!",nyquist,minVal);	'warn user
            else
                Message("The Edge frequencies must be less than %01.2f (1/2 the sample rate)\nand greater than %f (1/5000 the sample rate) for the selected data!",nyquist,minVal);	'warn user
            endif;    
           ok%:=0;		'Reset ok% value so that dialog repeats
        endif;
                   
	endif;	
until ok%>0;		'until we have acceptable settings to proceed
return ok%;
end;


Func DlgSettings%(item%)'This function monitors user changes to the dialog so that only relevant fields can be selected

v%:= DlgValue(2);		'Get type of filter
d%:= DlgValue(3);		'Check if Notch selected
f%:= DlgValue(4);		'Or Resonator...
c%:= DlgValue(1);		'Get filter name

DlgEnable(0,8,9);

docase
	case d%=1 then			    'If Notch filter selected
		DlgValue(4,0);		    'Clear resonator
		DlgValue(3,1);		    'Set Notch
		DlgEnable(0,1,2,5,7,8);	'Set dialog to initial state
		DlgEnable(1,10,11);

	case f%=1 then			    'If Resonator selected
		DlgValue(3,0);		    'Clear Notch
		DlgValue(4,1);		    'Set Resonator
		DlgEnable(0,1,2,5,7,8);	'Set dialog to initial state
		DlgEnable(1,10,11);

	case d%=0 and f%=0 then		'If neither selected, enable all standard fields
		DlgEnable(1,1,2,5,7);	'Enable Filter name field
		DlgEnable(0,10,11);		'Disable Notch.. & Resonator settings
		
		if v%=2 or v%=3  then	'If Bandpass or Bandstop filter selected..
			DlgEnable(1,8);		'..Enable Edge frequency 2 field
		endif;
		if c%=2 or c%=3 then	'If either Chebyshev filter selected..
			DlgEnable(1,9);		'..Enable ripple field
	 	endif;
endcase;
return 1;
end;

Func SetChans%(); 

DlgCreate("Additional channels for filtering");
for c%:=1 to useChans%[0] do
    DlgCheck(c%,chanNum$[c%]+chanNames$[c%],1,1.5+c%);
    next;
DlgText("Select up to 3 additional channels to filter\nThey must match the sample rate of the source channel",1,1);
ok%:= DlgShow(checkChans%[]);
if ok% < 0 then
    ArrConst(checkChans%[],0);
    endif;
return 1;
end;

Func IIRMake%()

docase	
	case notch%=1 then	
		IIRNotch(0,nFreq,qFact);		'Generate Notch filter
        filtMessage$:=Print$("IIR Notch filter with a frequency of %01.2fHz and q-factor of %d",freq,qFact);
        type%:=1;
        Resize save[useCh%+1][6];              'Set filter save array size for number of chnanels and filter type
    
	case reson%=1 then	
		IIRReson(0,nFreq,qFact);		'Generate Resonator filter
        filtMessage$:= Print$("IIR Resonator filter with a frequency of %01.2fHz and q-factor of %d",freq,qFact);
        type%:=2;
        Resize save[useCh%+1][6];
    
	case filtType%=0 then	
		if filtName%>=2 then		'If Chebyshev type filter..
			IIRLp(0,cFreq1,order%,filtName%,ripple);	'..generate Lowpass filter including ripple setting
			filtMessage$:=Print$("%s Lowpass filter with order of %d, edge frequency of %01.2fHz and ripple of %ddB",filtList$[filtName%],order%,eFreq1,ripple);
            
        else
			IIRLp(0,cFreq1,order%,filtName%);	'Otherwise generate Lowpass filter
			filtMessage$:=Print$("%s Lowpass filter with order of %d and edge frequency of %01.2fHz",filtList$[filtName%],order%,eFreq1); 
            
        endif;
        type%:=3;
        Resize save[useCh%+1][(order% * 2) + 2];     
	case filtType%=1 then	
		if filtName%>=2 then		'If Chebyshev type filter..
			IIRHp(0,cFreq1,order%,filtName%,ripple);	'..generate Highpass filter including ripple setting
			filtMessage$:=Print$("%s Highpass filter with order of %d, edge frequency of %01.2fHz and ripple of %ddB",filtList$[filtName%],order%,eFreq1,ripple);
            
        else
			IIRHp(0,cFreq1,order%,filtName%);	'Otherwise generate Highpass filter
			filtMessage$:=Print$("%s Highpass filter with order of %d and edge frequency of %01.2fHz",filtList$[filtName%],order%,eFreq1);
            
        endif;
        type%:=4;
        Resize save[useCh%+1][(order% * 2) + 2];  

	case filtType%=2 then	
		if filtName%>=2 then		'If Chebyshev type filter..
			IIRBp(0,cFreq1,cFreq2,order%,filtName%,ripple);	'..generate Bandpass filter including ripple setting
			filtMessage$:=Print$("%s Bandpass filter with order of %d, edge frequencies of %01.2fHz and %01.2fHz, and ripple of %ddB",filtList$[filtName%],order%,eFreq1,eFreq2,ripple);
            
        else
			IIRBp(0,cFreq1,cFreq2,order%,filtName%);	'Otherwise generate Bandpass filter
			filtMessage$:=Print$("%s Bandpass filter with order of %d and edge frequencies of %01.2fHz and %01.2fHz",filtList$[filtName%],order%,eFreq1,eFreq2); 
           
        endif;
        type%:=5;
        Resize save[useCh%+1][(order% * 4) + 2];

	case filtType%=3 then	
		if filtName%>=2 then		'If Chebyshev type filter.. 
			IIRBs(0,cFreq1,cFreq2,order%,filtName%,ripple);	'..generate Bandstop filter including ripple setting
			filtMessage$:=Print$("%s Bandstop filter with order of %d, edge frequencies of %01.2fHz and %01.2fHz, and ripple of %ddB",filtList$[filtName%],order%,eFreq1,eFreq2,ripple);
            
        else
			IIRBs(0,cFreq1,cFreq2,order%,filtName%);	'Otherwise generate Bandstop filter
			filtMessage$:=Print$("%s Bandstop filter with order of %d and edge frequencies of %01.2fHz and %01.2fHz",filtList$[filtName%],order%,eFreq1,eFreq2);
            
        endif;
    type%:=6;
    Resize save[useCh%+1][(order% * 4) + 2];
endcase;
return type%;
end;

proc DoToolbar()
ToolbarSet(1,"&Quit",Quit%);		'Set up toolbar buttons
ToolbarSet(2,"&Sample start", Start%);
ToolbarSet(3,"&Sample stop", Stop%);
ToolbarSet(4,"&New file", New%);
ToolbarEnable(3,0);					'Disable "Sample stop" button
Toolbar("Press SAMPLE START to commence sampling", 1023);'Wait here until quit
end;

func Quit%()							'If "Quit" is pressed
SampleStop();							'Stop sampling
return 0;								'leave toolbar
end;

func Start%()							'If "Start" is pressed
ToolbarSet(0,"",Idle%);				'Call Idle%() whenever there is free time
sTime := 0;                      'set start of anaylsis time
SampleStart();							'Start sampling
ToolbarEnable(4,0);					'Disable "New file" button
ToolbarEnable(3,1);					'Enable "Sample stop" button
ToolbarEnable(2,0);					'Disable "Sample start" button
ToolbarEnable(1,0);					'Disable "Quit" button
ToolbarText("Press SAMPLE STOP to stop sampling");
return 1;								'Stay with toolbar
end;

func Stop%()							'If "Stop" is pressed
SampleStop();							'Stop sampling
if (data%>0) and                 'if no data in file, stop will close it
   (ViewKind(data%) <> 0) then data% := 0 endif;
ToolbarEnable(4,1);					'Enable "New file" button
ToolbarEnable(3,0);					'Disable "Sample stop" button
ToolbarEnable(1,1);					'Enable "Quit" button
ToolbarClear(0);
ToolbarText("Press FILE NEW to capture more data");
return 1;								'Stay in toolbar
end;

func Idle%()							'Idle routine runs when system has time
var eTime;
var pts%;
var data[rate];
var i%;
var fTime;

if ViewKind(data%)=0 then        'This checks that the data view exists...
   View(data%);                  '...as this gives an error if it doesn't
   eTime:=Maxtime();

    if (eTime > sTime+BinSize(chans%[1])) then
        for i% := 1 to chans%[0] do
            'for i% := 1 to 1 do
            pts%:=ChanData(chans%[i%],data[],sTime,eTime,fTime);
        if pts% > 0 then
            Apply%(useFilt%,data[:pts%],i%);       'Apply selected filter to the incoming data
            ChanWriteWave(filtCh%[i%-1],data[:pts%],fTime);
            endif;
        next;
      sTime:=eTime;
      endif;
   endif;
return 1;								'Stay in toolbar
end;



Func Apply%(which%,data[],n%);                       'Function to apply filter
docase
    case which%=1 then                               'Apply type of filter selected to curent channel and save settings for next time this channel filtered
        IIRNotch(data[],2,save[n%][]);
    case which%=2 then
        IIRReson(data[],2,save[n%][]);
    case which%=3 then
        IIRLp(data[],2,save[n%][]);
    case which%=4 then
        IIRHp(data[],2,save[n%][]);
    case which%=5 then
        IIRBp(data[],2,save[n%][]);
    case which%=6 then
        IIRBs(data[],2,save[n%][]);
endcase;    
return 1;
end;

proc HideAll()
var i%;
gFloat%[0] := App(-2);        'number of windows
for i% := 1 to gFloat%[0] do  'hide all windows and save state
   gFloat%[i%] := View(App(i%)).WindowVisible(0);
   next;
end

proc RestoreAll()
var i%;
for i% := 1 to gFloat%[0] do  'restore hidden windows
   View(App(i%)).WindowVisible(gFloat%[i%]);
   next;
end

'$Ctrl200series.sgs|Example script to provide basic control over Magstim 200 series models that have been
'fitted with the parallel port interface option.

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.


var data%;                      'View handle of data file
var sRate% := 5000;             'Sample rate of waveform channels
var fLength := 500;             'Frame length (ms);
var chargeTime := 4;            'Additional outputs frame time for charging Magstim
var sPts%;                      'Data points per frame
var sOffset% := 100;            'Pre trigger time (ms)
var sOffset;                    'Pre trigger time in seconds
var nStates% := 5;              'Number of additional states
var int%[20];                   'Array of Magstim intensity levels (one per state)
var label$[20];                 'Array of state labels  
var FrameInt[20];               'Array in intervals to next frame
var TrigDir% := 1;              'Trigger polarity

ArrConst(frameInt[],5);         'Initialise frame interval

ToolbarSet(1, "Quit");          'Toolbar for script control
ToolbarSet(3, "Settings", GetSet%);
ToolbarSet(4, "New file", New%);
ToolbarEnable(4,0);
Toolbar("TMS configuration setup",1023);


Func GetSet%();
var ok%;
var okState%;
DlgCreate("Sampling parameters");               'Set sampling parameters 
DlgInteger(1, "Sample rate",100,100000);
DlgReal(2, "Frame length(ms)",50,10000);
DlgInteger(3, "Pre-trigger time",0,500);
DlgReal(4, "Charge time",0,10);
DlgInteger(5, "Number of states",1,20);
DlgList(6, "Trigger polarity","Rising|Falling",2);
ok% := DlgShow(sRate%,fLength,sOffset%,chargeTime,nStates%,trigDir%);           'Assign return values to variables
if ok% > 0 then                                                                 'If OK pressed        
    sPts% := (fLength/1000.0)* sRate%;                                          'Get number of sample points per frame
    sOffset := sOffset% / 1000.0;                                               'Convert offset to seconds
    okState% := GetStateParams%(NStates%);                                      'Jump to function to get individual state settings
    if okState% > 0 then
        SetConfig%();                                                           'Set up configuration
        ToolbarEnable(4,1);
    endif;
endif;
return 1;
end;


'Function to get & set individual state parameters (Magstim intensity, state label and interval to next frame)
Func GetStateParams%(n%);                               
var ok%;
var i%;
DlgCreate("State setup");       'Setup Magstim settings for multiple states
DlgText("State",0,1);
DlgText("Intensity",12,1);
DlgText("Label",26,1);
DlgText("Interval(s)",36,1);
for i% := 1 to n% do
    DlgText(Str$(i%),3,i%+1);
    DlgInteger(i%,3,1,127,14,i%+1);
    DlgString(n%+i%,"",10,"",24,i%+1);
    DlgReal((n%*2)+i%,5,chargeTime + 0.1,3600,40,i%+1);
next;
ok% := DlgShow(int%[:nStates%],label$[:nStates%],frameInt[:nStates%]);
if ok% > 0 then
    if trigDir% = 1 then
        ArrAdd(int%[:nStates%],128);        
    endif;
endif;
return ok%;
end;


Func New%()
data% := FileNew(0,0);          'Open new file for sampling
if data% > 0 then
    Window(0,0,100,100);                
    FrontView(data%);
    ToolbarClear(4);
    ToolbarSet(4, "Start",Start%);
endif;
return 1;
end;


Func Start%()
View(data%);
SampleStart();      'Start sampling
ToolbarClear(4);
ToolbarSet(4, "Stop",Stop%);
return 1;
end;


Func Stop%()
View(data%);
SampleStop();       'Stop sampling
ToolbarClear(4);
ToolbarSet(4, "New file", New%);
return 1
end;


Func SetConfig%()
var i%;
SampleClear();                                      'Clear existing configuration
SampleBurst(1);                                     'Burst mode sampling
SamplePorts(2);                                     'Sample 2 ports
SamplePortUnits$(0," Volt");                        
SamplePortFull(0,5);
SamplePortZero(0,0);
SamplePortName$(0,"ADC 0");
SamplePortUnits$(1," Volt");
SamplePortFull(1,5);
SamplePortZero(1,0);
SamplePortName$(1,"ADC 1");
SampleKeyMark(1);                                   'Enable keyboard marker
SampleDigMark(0);                                   'Disable digital marker
SampleAutoName$("");                                'No auto file naming
SampleAutoFile(0);              
SamplePoints(sPts%);                                'Number of data points per ADC port per frame
SampleRate(sRate%);                                 'Sample rate
SampleZeroOffset(sOffset);                          'Pre zero time
SampleTrigger(1);                                   'Turn on sweep trigger
SampleMode(3);                                      'Sample mode to fixed interval
SampleOutMode(1);                                   'Pulses outut mode
SampleDigOMask(255);                                'Enable all 8 digital outputs
SampleOutClock(0.001);                              'Set output resolution to 1ms
SampleStates(nStates%);                             'Number of states
SampleStatesMode(2);                                'Dynamic outputs state mode
SampleStatesOrder(0);                               'Numeric state ordering
SampleStatesIdle(0);
SampleStatesRepeats(1);
SampleStatesOptions(2);
SamplePause(0);
SampleWrite(1);
PulseClear(0,-1);                                                  'Clear all pulses
if trigDir% = 0 then
    PulseDataSet(0,-1,0,0);
else
    PulseDataSet(0,-1,0,128);
endif;
for i% := 1 to nStates% do                                              'Set up individual state parameters
    PulseClear(i%,-1); 
    SampleOutLength(i%,chargeTime+(fLength/1000.0));                    'Set output length to same as frame length
    SampleOutTrig(i%,chargeTime);                                       'Set outputs frame to include charge time
    PulseDataSet(i%,-1,0,int%[i%-1]);                                   'Set bit value to intensity level 
    PulseAdd(i%,-1,1,"Stimulus",chargeTime + sOffset,0.01);             'Add stimulus trigger pulse at time zero
    PulseDataSet(i%,-1,"Stimulus",128);                             'Set it to be on digital output 7 
    SampleStateLabel$(i%,label$[i%-1]);                                 'Set state label
    SampleFixedInt(i%,frameInt[i%-1]);                                  'Set interval to next frame
next;
return 1;
end;

each state used. The connection between the Magstim parallel port interface and 
the 1401 digital outputs can be achieved with a modified cable, available from 
CED.   

Did you know…? 
You can play a waveform channel as audio through a sound card from the offline Output 
waveform dialog in Spike2, which is available from the Sample menu. If there is a sound 
card available on the system, you can select Sound instead of a DAC from the Output 
waveform dialog. 

Recent questions   
Q. I would like to be able to use the 1401 to trigger both pulses of a BiStim unit so 

that I can set the delay in Signal, rather than have the BiStim trigger the second 
pulse. 

A. By default the first output pulse occurs at the time of the trigger and the second 
at a preset interval after the trigger, the interval being internally timed by the 
BiStim unit itself. It is however possible to use the 1401 digital outputs to trigger 
both the main pulse and the second (Power B) pulse by setting the BiStim to use 
independent trigger mode on the front panel of the master unit (this is referred to 
as IBT mode in the BiStim documentation). Once IBT mode has been set on the 
BiStim unit, select the associated checkbox in the Magstim configuration dialog in 
Signal. 

 
Magstim configuration options for BiStim units 

With these settings enabled both the Power and Power B stimuli are triggered in 
response to a TTL input to the BiStim master unit. 

User group      
You can now find our eNewsletters archived on the CED website. 

Try the CED Forums bulletin board for software and hardware support 

If you have any comments about the newsletter format and content, or wish to un-
subscribe from the mailing list for this newsletter, please notify sales@ced.co.uk. 
All Trademarks are acknowledged to be the Trademarks of the registered holders. 

